

Análisis Sintáctico LL(1)

Parser Mini-0

Piero Omar De La Cruz Mancilla
Enyelbert Anderson Panta Huaracha

`pdelacruz@ulasalle.edu.pe`
`epantah@ulasalle.edu.pe`

Universidad La Salle
Carrera Profesional de Ingeniería de Software
Curso: Compiladores

28 de noviembre de 2025

Índice

| | |
|---|-----------|
| 1. Introducción | 3 |
| 2. Gramática Formal Transformada | 3 |
| 2.1. Notación Utilizada | 3 |
| 2.2. Producciones Completas (70 producciones) | 3 |
| 2.2.1. Programa y Declaraciones (1-14) | 3 |
| 2.2.2. Sentencias (15-33) | 3 |
| 2.2.3. Expresiones con Precedencia (34-70) | 4 |
| 3. Conjuntos FIRST | 5 |
| 3.1. Definición | 5 |
| 3.2. Conjuntos FIRST Calculados | 5 |
| 4. Conjuntos FOLLOW | 6 |
| 4.1. Definición | 6 |
| 4.2. Conjuntos FOLLOW Calculados | 6 |
| 5. Tabla de Análisis Sintáctico LL(1) | 7 |
| 5.1. Introducción | 7 |
| 5.2. Construcción de la Tabla | 7 |
| 5.3. Tabla Principal (Extracto) | 7 |
| 5.4. Entradas por Sección | 7 |
| 6. Verificación de Propiedades LL(1) | 8 |
| 6.1. Condiciones LL(1) | 8 |
| 6.2. Verificación Caso por Caso | 8 |
| 6.3. Conclusión | 9 |
| 7. Resolución de Conflictos | 9 |
| 7.1. Conflicto 1: Declaración vs Asignación | 9 |
| 7.2. Conflicto 2: Operador = Dual | 9 |
| 7.3. Conflicto 3: Recursión Izquierda | 10 |
| 8. Correspondencia con la Implementación | 10 |
| 8.1. Mapeo Gramática-Código | 10 |
| 8.2. Sistema de Lookahead | 10 |
| 9. Ejemplo de Derivación | 10 |
| 9.1. Programa de Entrada | 10 |
| 9.2. Derivación por la Izquierda (simplificada) | 11 |
| 10. Conclusiones | 11 |
| 10.1. Resultados del Análisis | 11 |
| 10.2. Ventajas de la Gramática LL(1) | 11 |
| 10.3. Aplicaciones | 11 |
| 11. Referencias | 12 |

1. Introducción

Este documento presenta el análisis formal completo de la gramática LL(1) utilizada en la implementación del parser recursivo descendente para el lenguaje Mini-0. El análisis incluye:

- Gramática formal transformada (70 producciones)
- Conjuntos FIRST para todos los no-terminales
- Conjuntos FOLLOW para todos los no-terminales
- Tabla de análisis sintáctico LL(1)
- Verificación de propiedades LL(1)
- Resolución de conflictos y ambigüedades

2. Gramática Formal Transformada

2.1. Notación Utilizada

- **Terminales:** MAYÚSCULAS o entre comillas
- **No-terminales:** entre ángulos
- **Épsilon:** cadena vacía
- **— :** alternativa de producción

2.2. Producciones Completas (70 producciones)

2.2.1. Programa y Declaraciones (1-14)

- (1) `program` \rightarrow `decl-or-fun-list`
- (2) `decl-or-fun-list` \rightarrow `decl-or-fun decl-or-fun-list`
- (3) `decl-or-fun-list` $\rightarrow \varepsilon$
- (4) `decl-or-fun` \rightarrow `FUN ID (param-list) : type NL* stmt-list END NL*`
- (5) `decl-or-fun` \rightarrow `ID : type NL*`
- (6) `param-list` \rightarrow `param param-list'`
- (7) `param-list` $\rightarrow \varepsilon$
- (8) `param-list'` \rightarrow `, param param-list'`
- (9) `param-list'` $\rightarrow \varepsilon$
- (10) `param` \rightarrow `ID : type`
- (11) `type` \rightarrow `array-prefix base-type`
- (12) `array-prefix` \rightarrow `[] array-prefix`
- (13) `array-prefix` $\rightarrow \varepsilon$
- (14) `base-type` \rightarrow `INT — BOOL — STRING — CHAR`

2.2.2. Sentencias (15-33)

- (15) `stmt-list` \rightarrow `statement stmt-list`
- (16) `stmt-list` $\rightarrow \varepsilon$
- (17) `statement` \rightarrow `ID : type NL*`
- (18) `statement` \rightarrow `ID assign-or-call NL*`
- (19) `statement` \rightarrow `RETURN expr NL*`
- (20) `statement` \rightarrow `IF expr NL* stmt-list else-part END NL*`
- (21) `statement` \rightarrow `WHILE expr NL* stmt-list LOOP NL*`
- (22) `statement` \rightarrow `expr NL*`

- (23) assign-or-call \rightarrow = expr
- (24) assign-or-call \rightarrow [expr] = expr
- (25) assign-or-call \rightarrow (arg-list) index-chain
- (26) else-part \rightarrow ELSE NL* stmt-list
- (27) else-part $\rightarrow \varepsilon$
- (28) arg-list \rightarrow expr arg-list'
- (29) arg-list $\rightarrow \varepsilon$
- (30) arg-list' \rightarrow , expr arg-list'
- (31) arg-list' $\rightarrow \varepsilon$
- (32) index-chain \rightarrow [expr] index-chain
- (33) index-chain $\rightarrow \varepsilon$

2.2.3. Expresiones con Precedencia (34-70)

Nivel 1: OR (34-37)

- (34) expr \rightarrow or-expr
- (35) or-expr \rightarrow and-expr or-expr'
- (36) or-expr' \rightarrow OR and-expr or-expr'
- (37) or-expr' $\rightarrow \varepsilon$

Nivel 2: AND (38-40)

- (38) and-expr \rightarrow equality-expr and-expr'
- (39) and-expr' \rightarrow AND equality-expr and-expr'
- (40) and-expr' $\rightarrow \varepsilon$

Nivel 3: Igualdad (41-44)

- (41) equality-expr \rightarrow relational-expr equality-expr'
- (42) equality-expr' \rightarrow = relational-expr equality-expr'
- (43) equality-expr' \rightarrow <> relational-expr equality-expr'
- (44) equality-expr' $\rightarrow \varepsilon$

Nivel 4: Relacional (45-50)

- (45) relational-expr \rightarrow additive-expr relational-expr'
- (46) relational-expr' \rightarrow > additive-expr relational-expr'
- (47) relational-expr' \rightarrow < additive-expr relational-expr'
- (48) relational-expr' \rightarrow >= additive-expr relational-expr'
- (49) relational-expr' \rightarrow <= additive-expr relational-expr'
- (50) relational-expr' $\rightarrow \varepsilon$

Nivel 5: Aditivo (51-54)

- (51) additive-expr \rightarrow multiplicative-expr additive-expr'
- (52) additive-expr' \rightarrow + multiplicative-expr additive-expr'
- (53) additive-expr' \rightarrow - multiplicative-expr additive-expr'
- (54) additive-expr' $\rightarrow \varepsilon$

Nivel 6: Multiplicativo (55-58)

- (55) multiplicative-expr \rightarrow unary-expr multiplicative-expr'
- (56) multiplicative-expr' \rightarrow * unary-expr multiplicative-expr'
- (57) multiplicative-expr' \rightarrow / unary-expr multiplicative-expr'
- (58) multiplicative-expr' $\rightarrow \varepsilon$

Nivel 7: Unario (59-61)

- (59) unary-expr \rightarrow - unary-expr
- (60) unary-expr \rightarrow NOT unary-expr
- (61) unary-expr \rightarrow primary-expr

Nivel 8: Primarias (62-70)

- (62) primary-expr \rightarrow LITNUMERAL
- (63) primary-expr \rightarrow LITSTRING
- (64) primary-expr \rightarrow TRUE
- (65) primary-expr \rightarrow FALSE

- (66) $\text{primary-expr} \rightarrow \text{ID postfix-chain}$
- (67) $\text{primary-expr} \rightarrow (\text{expr})$
- (68) $\text{postfix-chain} \rightarrow (\text{arg-list}) \text{postfix-chain}$
- (69) $\text{postfix-chain} \rightarrow [\text{expr}] \text{postfix-chain}$
- (70) $\text{postfix-chain} \rightarrow \varepsilon$

3. Conjuntos FIRST

3.1. Definición

Para un símbolo gramatical alfa, $\text{FIRST}(\text{alfa})$ es el conjunto de terminales que pueden aparecer como primer símbolo en cadenas derivadas de alfa.

3.2. Conjuntos FIRST Calculados

- $\text{FIRST}(\text{program}) = \{\text{FUN}, \text{ID}, \text{épsilon}\}$
- $\text{FIRST}(\text{decl-or-fun-list}) = \{\text{FUN}, \text{ID}, \text{épsilon}\}$
- $\text{FIRST}(\text{decl-or-fun}) = \{\text{FUN}, \text{ID}\}$
- $\text{FIRST}(\text{param-list}) = \{\text{ID}, \text{épsilon}\}$
- $\text{FIRST}(\text{param-list}') = \{\text{comma}, \text{épsilon}\}$
- $\text{FIRST}(\text{param}) = \{\text{ID}\}$
- $\text{FIRST}(\text{type}) = \{\text{lbracket}, \text{INT}, \text{BOOL}, \text{STRING}, \text{CHAR}\}$
- $\text{FIRST}(\text{array-prefix}) = \{\text{lbracket}, \text{épsilon}\}$
- $\text{FIRST}(\text{base-type}) = \{\text{INT}, \text{BOOL}, \text{STRING}, \text{CHAR}\}$
- $\text{FIRST}(\text{stmt-list}) = \{\text{ID}, \text{RETURN}, \text{IF}, \text{WHILE}, \text{lparen}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}, \text{épsilon}\}$
- $\text{FIRST}(\text{statement}) = \{\text{ID}, \text{RETURN}, \text{IF}, \text{WHILE}, \text{lparen}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{assign-or-call}) = \{\text{equals}, \text{lbracket}, \text{lparen}\}$
- $\text{FIRST}(\text{else-part}) = \{\text{ELSE}, \text{épsilon}\}$
- $\text{FIRST}(\text{expr}) = \{\text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{or-expr}) = \{\text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{or-expr}') = \{\text{OR}, \text{épsilon}\}$
- $\text{FIRST}(\text{and-expr}) = \{\text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{and-expr}') = \{\text{AND}, \text{épsilon}\}$
- $\text{FIRST}(\text{equality-expr}) = \{\text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{equality-expr}') = \{\text{equals}, \text{notequal}, \text{épsilon}\}$
- $\text{FIRST}(\text{relational-expr}) = \{\text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{relational-expr}') = \{\text{gt}, \text{lt}, \text{gte}, \text{lte}, \text{épsilon}\}$
- $\text{FIRST}(\text{additive-expr}) = \{\text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$

- $\text{FIRST}(\text{additive-expr}') = \{\text{plus}, \text{minus}, \text{épsilon}\}$
- $\text{FIRST}(\text{multiplicative-expr}) = \{\text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{multiplicative-expr}') = \{\text{star}, \text{slash}, \text{épsilon}\}$
- $\text{FIRST}(\text{unary-expr}) = \{\text{minus}, \text{NOT}, \text{lparen}, \text{ID}, \text{LITNUMERAL}, \text{LITSTRING}, \text{TRUE}, \text{FALSE}\}$
- $\text{FIRST}(\text{primary-expr}) = \{\text{LITNUMERAL}, \text{LITSTRING}, \text{TRUE}, \text{FALSE}, \text{ID}, \text{lparen}\}$
- $\text{FIRST}(\text{postfix-chain}) = \{\text{lparen}, \text{lbracket}, \text{épsilon}\}$

4. Conjuntos FOLLOW

4.1. Definición

Para un no-terminal A, $\text{FOLLOW}(A)$ es el conjunto de terminales que pueden aparecer inmediatamente después de A en alguna forma sentencial.

4.2. Conjuntos FOLLOW Calculados

- $\text{FOLLOW}(\text{program}) = \{\text{dollar}\}$
- $\text{FOLLOW}(\text{decl-or-fun-list}) = \{\text{dollar}\}$
- $\text{FOLLOW}(\text{decl-or-fun}) = \{\text{FUN}, \text{ID}, \text{dollar}\}$
- $\text{FOLLOW}(\text{param-list}) = \{\text{rparen}\}$
- $\text{FOLLOW}(\text{param-list}') = \{\text{rparen}\}$
- $\text{FOLLOW}(\text{param}) = \{\text{comma}, \text{rparen}\}$
- $\text{FOLLOW}(\text{type}) = \{\text{NL}, \text{comma}, \text{rparen}\}$
- $\text{FOLLOW}(\text{array-prefix}) = \{\text{INT}, \text{BOOL}, \text{STRING}, \text{CHAR}\}$
- $\text{FOLLOW}(\text{base-type}) = \{\text{NL}, \text{comma}, \text{rparen}\}$
- $\text{FOLLOW}(\text{stmt-list}) = \{\text{END}, \text{ELSE}, \text{LOOP}\}$
- $\text{FOLLOW}(\text{statement}) = \{\text{ID}, \text{RETURN}, \text{IF}, \text{WHILE}, \text{END}, \text{ELSE}, \text{LOOP}, \text{lparen}, \text{LITNUMERAL}, \text{LITSTRING}, \text{minus}, \text{NOT}, \text{TRUE}, \text{FALSE}\}$
- $\text{FOLLOW}(\text{assign-or-call}) = \{\text{NL}\}$
- $\text{FOLLOW}(\text{else-part}) = \{\text{END}\}$
- $\text{FOLLOW}(\text{arg-list}) = \{\text{rparen}\}$
- $\text{FOLLOW}(\text{arg-list}') = \{\text{rparen}\}$
- $\text{FOLLOW}(\text{index-chain}) = \{\text{NL}, \text{comma}, \text{rparen}, \text{rbracket}, \text{AND}, \text{OR}, \text{equals}, \text{notequal}, \text{gt}, \text{lt}, \text{gte}, \text{lte}, \text{plus}, \text{minus}, \text{star}, \text{slash}\}$
- $\text{FOLLOW}(\text{expr}) = \{\text{NL}, \text{comma}, \text{rparen}, \text{rbracket}\}$
- $\text{FOLLOW}(\text{or-expr}) = \{\text{NL}, \text{comma}, \text{rparen}, \text{rbracket}\}$
- $\text{FOLLOW}(\text{or-expr}') = \{\text{NL}, \text{comma}, \text{rparen}, \text{rbracket}\}$
- $\text{FOLLOW}(\text{and-expr}) = \{\text{OR}, \text{NL}, \text{comma}, \text{rparen}, \text{rbracket}\}$

- FOLLOW(and-expr') = {OR, NL, comma, rparen, rbracket}
- FOLLOW(equality-expr) = {AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(equality-expr') = {AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(relational-expr) = {equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(relational-expr') = {equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(additive-expr) = {gt, lt, gte, lte, equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(additive-expr') = {gt, lt, gte, lte, equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(multiplicative-expr) = {plus, minus, gt, lt, gte, lte, equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(multiplicative-expr') = {plus, minus, gt, lt, gte, lte, equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(unary-expr) = {star, slash, plus, minus, gt, lt, gte, lte, equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(primary-expr) = {star, slash, plus, minus, gt, lt, gte, lte, equals, notequal, AND, OR, NL, comma, rparen, rbracket}
- FOLLOW(postfix-chain) = {star, slash, plus, minus, gt, lt, gte, lte, equals, notequal, AND, OR, NL, comma, rparen, rbracket}

5. Tabla de Análisis Sintáctico LL(1)

5.1. Introducción

La tabla LL(1) es una matriz $M[A, a]$ donde A es un no-terminal, a es un terminal o dollar, y $M[A, a]$ contiene la producción a aplicar cuando el parser está en el estado A y ve el token a.

5.2. Construcción de la Tabla

Para cada producción A produce alfa:

1. Para cada terminal a en FIRST(alfa), añadir A produce alfa a $M[A, a]$
2. Si épsilon está en FIRST(alfa), para cada terminal b en FOLLOW(A), añadir A produce alfa a $M[A, b]$

5.3. Tabla Principal (Extracto)

| tableheader | | | | | | |
|-------------|-----|----------|------|------|------|-----|
| program | (1) | (1) | - | - | - | (1) |
| decl-or-fun | (4) | (5) | - | - | - | - |
| type | - | - | (11) | - | - | - |
| statement | - | (17)(18) | - | (20) | (19) | - |
| expr | - | (34) | - | - | - | - |

Cuadro 1: Tabla LL(1) - Sección principal

5.4. Entradas por Sección

Declaraciones:

- $M[\text{decl-or-fun}, \text{FUN}] = (4)$
- $M[\text{decl-or-fun}, \text{ID}] = (5)$

Tipos:

- $M[\text{type}, \text{lbracket}] = (11)$
- $M[\text{type}, \text{INT}] = (11)$
- $M[\text{type}, \text{BOOL}] = (11)$
- $M[\text{type}, \text{STRING}] = (11)$
- $M[\text{type}, \text{CHAR}] = (11)$

Sentencias:

- $M[\text{statement}, \text{ID}] = (17)$ o (18) según lookahead
- $M[\text{statement}, \text{RETURN}] = (19)$
- $M[\text{statement}, \text{IF}] = (20)$
- $M[\text{statement}, \text{WHILE}] = (21)$

Expresiones:

- $M[\text{expr}, t] = (34)$ para todo t en $\text{FIRST}(\text{expr})$
- $M[\text{or-expr}', \text{OR}] = (36)$
- $M[\text{or-expr}', t] = (37)$ para todo t en $\text{FOLLOW}(\text{or-expr}')$
- $M[\text{and-expr}', \text{AND}] = (39)$
- $M[\text{and-expr}', t] = (40)$ para todo t en $\text{FOLLOW}(\text{and-expr}')$

6. Verificación de Propiedades LL(1)

6.1. Condiciones LL(1)

Una gramática es LL(1) si y solo si para cada no-terminal A con producciones $A \rightarrow \alpha_1 \mid \alpha_2 \mid \dots \mid \alpha_N$:

1. $\text{FIRST}(\alpha_i) \cap \text{FIRST}(\alpha_j) = \emptyset$ para todo i distinto de j
2. Si ϵ está en $\text{FIRST}(\alpha_i)$, entonces $\text{FIRST}(\alpha_j) \cap \text{FOLLOW}(A) = \emptyset$ para todo j distinto de i

6.2. Verificación Caso por Caso

Caso 1: decl-or-fun

Producciones: (4) $\text{FUN} \dots$ y (5) $\text{ID} : \dots$

Verificación:

- $\text{FIRST}(\text{producción 4}) = \{\text{FUN}\}$
- $\text{FIRST}(\text{producción 5}) = \{\text{ID}\}$
- Intersección = vacío (correcto)

Caso 2: statement (ambigüedad con ID)

Producciones: (17) ID : type y (18) ID assign-or-call

Se resuelve con lookahead de 1 token:

- Si peek_token(1) == colon entonces producción (17)
- Si peek_token(1) == equals o lbracket entonces producción (18)

Caso 3: Expresiones con primos

Para todas las producciones con primos (or-expr', and-expr', etc.):

- FIRST(operador) = operador específico
- FIRST(épsilon) = épsilon
- Intersección = vacío (correcto)
- FIRST(operador) intersección FOLLOW(A') = vacío (correcto)

6.3. Conclusión

La gramática es LL(1) válida. Todas las producciones cumplen las condiciones requeridas.

7. Resolución de Conflictos**7.1. Conflicto 1: Declaración vs Asignación**

Problema:

```
1 x: int      // Declaracion
2 x = 10     // Asignacion
```

Ambas construcciones comienzan con ID.

Solución: Lookahead de 1 token con peek_token.

```
1 if (cur_token()->tipo == TK_ID &&
2     peek_token(1)->tipo == TK_COLON) {
3     parse_declaration();
4 } else if (cur_token()->tipo == TK_ID &&
5           (peek_token(1)->tipo == TK_EQ ||
6            peek_token(1)->tipo == TK_LBRACKET)) {
7     parse_assignment();
8 }
```

7.2. Conflicto 2: Operador = Dual

Problema: El símbolo = tiene dos significados:

- En contexto de sentencia: operador de asignación
- En contexto de expresión: operador de comparación

```
1 x = 10      // Asignacion (sentencia)
2 if x = y    // Comparacion (expresion)
```

Solución: Separación sintáctica por contexto. La gramática distingue explícitamente estos contextos mediante producciones separadas:

- Producción (23): asignación en sentencias
- Producción (42): comparación en expresiones

7.3. Conflicto 3: Recursión Izquierda

Problema: E produce $E + T \mid E - T \mid T$ (recursión izquierda)

Solución: Transformación con primos:

E produce $T E'$

E' produce $+ T E' \mid - T E' \mid \text{épsilon}$

Esta transformación se aplicó a todos los niveles de precedencia de operadores.

8. Correspondencia con la Implementación

8.1. Mapeo Gramática-Código

| tableheader | |
|---------------------|------------------------|
| program | parse_program() |
| type | parse_type() |
| statement | parse_statement() |
| expr | parse_expression() |
| or-expr | parse_or_expr() |
| and-expr | parse_and_expr() |
| equality-expr | parse_equality() |
| relational-expr | parse_relational() |
| additive-expr | parse_additive() |
| multiplicative-expr | parse_multiplicative() |
| unary-expr | parse_unary() |
| primary-expr | parse_primary() |

Cuadro 2: Correspondencia entre gramática e implementación

8.2. Sistema de Lookahead

El parser utiliza dos funciones auxiliares para inspeccionar tokens:

```

1 // Token actual
2 static Token *cur_token() {
3     if (pos >= num_tokens) return NULL;
4     return tokens[pos];
5 }
6
7 // Token a distancia 'offset' del actual
8 static Token *peek_token(int offset) {
9     int idx = pos + offset;
10    if (idx >= num_tokens) return NULL;
11    return tokens[idx];
12 }

```

Esto permite implementar lookahead de 1 token ($k=1$) requerido por LL(1).

9. Ejemplo de Derivación

9.1. Programa de Entrada

```

1 fun main(): int
2     x: int
3     x = 5 + 3
4     return x
5 end

```

9.2. Derivación por la Izquierda (simplificada)

```
program deriva decl-or-fun-list
deriva decl-or-fun decl-or-fun-list
deriva FUN main ( ) : INT stmt-list END épsilon
deriva FUN main ( ) : INT statement stmt-list END
deriva FUN main ( ) : INT x : INT stmt-list END
deriva FUN main ( ) : INT x : INT statement stmt-list END
deriva FUN main ( ) : INT x : INT x = 5 + 3 stmt-list END
deriva FUN main ( ) : INT x : INT x = 5 + 3 return x END
```

10. Conclusiones

10.1. Resultados del Análisis

1. La gramática transformada es LL(1) válida, cumpliendo todas las condiciones requeridas.
2. Se eliminó exitosamente toda recursión izquierda mediante transformación con producciones primas.
3. Las ambigüedades (declaración vs asignación, operador = dual) se resolvieron mediante lookahead y separación sintáctica.
4. La precedencia de operadores se implementó correctamente mediante estratificación de la gramática en 8 niveles.
5. Todos los conjuntos FIRST y FOLLOW están correctamente calculados y documentados.
6. La tabla LL(1) es completa y no contiene conflictos.
7. Existe correspondencia directa entre la gramática formal y la implementación en parser.c.

10.2. Ventajas de la Gramática LL(1)

- **Eficiencia:** Parsing en tiempo lineal $O(n)$
- **Determinismo:** Sin backtracking ni ambigüedades
- **Lookahead mínimo:** Solo requiere 1 token de anticipación
- **Implementación directa:** Cada no-terminal corresponde a una función
- **Manejo de errores:** Detección precisa de errores sintácticos

10.3. Aplicaciones

Este análisis formal sirve como base para:

- Validación teórica del parser implementado
- Documentación técnica completa del compilador
- Extensión futura del lenguaje Mini-0
- Referencia para implementaciones alternativas
- Material educativo sobre análisis sintáctico

11. Referencias

1. Aho, A. V., Lam, M. S., Sethi, R., & Ullman, J. D. (2006). *Compilers: Principles, Techniques, and Tools* (2nd ed.). Addison-Wesley.
2. Grune, D., van Reeuwijk, K., Bal, H. E., Jacobs, C. J. H., & Langendoen, K. (2012). *Modern Compiler Design* (2nd ed.). Springer.
3. Cooper, K. D., & Torczon, L. (2011). *Engineering a Compiler* (2nd ed.). Morgan Kaufmann.
4. Repositorio del proyecto: https://github.com/EnyelbertAnderson/Compiladores_Lab_10